

TD8 : Fonctions récursives et appels imbriqués

Objectif(s)

★ Ce TD a pour but de consolider les appels de fonctions et d'implémenter des fonctions récursives et imbriquées.

Note pédagogique:

Traiter une fonction récursive ainsi que la fonction `arimean` qui illustre bien aussi la nécessité de sauvegarder la valeur de l'argument avant d'effectuer les appels aux fonctions dans son corps.

Une fonction imbriquée est une fonction qui est appelée par une autre fonction. Les conventions d'appel sont faites pour pouvoir appeler des fonctions écrites soi-même ou par d'autres, sans avoir en avoir se soucier d'autre chose que de les avoir bien suivies.

Une fonction récursive est une fonction qui s'appelle elle-même. C'est donc une fonction qui appelle une fonction, cette fonction a juste le même nom : cela ne change rien au niveau des conventions d'appel ! C'est au programmeur de s'assurer que la fonction récursive termine, c'est un problème d'algorithmique pas de programmation assembleur.

Pour l'écriture de fonctions, vous suivrez consciencieusement les étapes de programmation d'une fonction en assembleur à savoir pour rappel (cf. l'énoncé de la semaine dernière aussi). Vous porterez une attention particulière au contenu des registres non persistants (par exemple §4) avant les appels de fonctions dans des fonctions et au besoin de sauvegarder la valeur des arguments dans certains cas (déterminez lesquels).

Exercice 1 – Fonction puissance linéaire

Question 1

On considère le code suivant :

```
int puissance(int x, int n) {
    int tmp;
    if (n == 0) {
        return 1;
    }
    else {
        tmp = puissance(x, n - 1);
        return x * tmp;
    }
}

void main() {
    int x = 3;
    int p = 2;

    printf("%d", puissance(x, p));
    printf("%d", puissance(2, 6));

    exit();
}
```

Donnez le code de la fonction puissance puis le code du programme principal correspondant au code C ci-dessus. N'oubliez pas qu'il faut allouer de l'espace mémoire sur la pile pour les variables locales, dans les fonctions mais aussi dans le programme principal (le main).

Solution:

```

1  .data
2
3  .text
4
5
6  main:
7      # x et p non optimises en registre
8      addiu $29, $29, -16 # na = 2 + nv = 2 donc 4 mots
9      ori $8, $0, 3      # initialisation de x
10     sw $8, 8($29)
11     ori $8, $0, 2      # initialisation de p
12     sw $8, 12($29)
13
14     # puissance(x,p)
15     lw $4, 8($29) # 1er arg = x
16     lw $5, 12($29) # 2eme arg = p
17     jal puissance
18     ori $4, $2, 0 # instruction d'adresse @a
19     ori $2, $0, 1 # affichage du resultat
20     syscall
21
22     # puissance(2,6)
23     ori $4, $0, 2 # 1er arg = 2
24     ori $5, $0, 6 # 2eme arg = 6
25     jal puissance
26     ori $4, $2, 0 # instruction d'adresse @b
27     ori $2, $0, 1
28     syscall
29
30     addiu $29, $29, 16
31     ori $2, $0, 10
32     syscall
33
34  puissance:
35     # Prologue
36     # On sauvegarde les 2 parametres meme si seul x doit obligatoirement etre sauvegarde
37     addiu $29, $29, -16 # nv = 1 + nr = 0 + 1 ($31) + na = 2 = 4 mots
38     sw $31, 12($29)
39     sw $4, 16($29) # sauvegarde x
40     sw $5, 20($29) # sauvegarde p
41
42     bne $5, $0, _else # si n!= cas recursif
43     ori $2, $0, 1 # sinon val retour = 1
44     j _fin_if
45  _else:
46     # $4 contient deja la bonne valeur
47     addiu $5, $5, -1 # p - 1
48     jal puissance
49     lw $8, 16($29) # relecture x ; instruction d'adresse @c
50     mult $8, $2 # res * x
51     mflo $2
52  _fin_if:
53     lw $31, 12($29)

```

```

54      addiu $29, $29, +16
55      jr     $31

```

Question 2 – Représentation graphique de l'arbre d'appels

Donnez les appels récursifs engendrés par l'appel `puissance(2, 6)`

Solution:

```

puissance(2, 6)
-> puissance(2, 5)
    -> puissance(2, 4)
        -> puissance(2, 3)
            -> puissance(2, 2)
                -> puissance(2, 1)
                    -> puissance(2, 0)

```

Question 3 – Représentation graphique de la pile

Numérotez les instructions de votre code assembleur.

Représentez graphiquement l'évolution de la pile au cours de l'exécution du premier appel à la fonction `puissance`. Vous indiquerez par `@i` les adresses de retour que vous mettrez sur la pile, avec `i` correspondant au numéro de l'instruction associée à cette adresse de retour.

Solution:

L'évolution de la pile lors de l'appel à `puissance(3, 2)` est représentée dans la Figure 1.




FIGURE 1 – Evolution de la pile lors d'un appel à `puiss(3, 2)`

Exercice 2 – Fonctions imbriquées

On souhaite écrire en assembleur MIPS32 le programme C qui calcule la valeur moyenne de tous les entiers (non négatifs) stockés dans un tableau de dimension quelconque. Par convention, et pour pouvoir réutiliser la fonction `sumtab()`, on suppose que le dernier élément du tableau possède une valeur négative. Ce dernier élément n'est pas pris en compte dans le calcul de la moyenne.

Le programme principal appelle la fonction `arimean()` qui a pour seul argument un pointeur sur un tableau d'entiers. Elle renvoie la valeur de la moyenne arithmétique des éléments du tableau (tronquée à la valeur entière inférieure). La fonction `arimean()` appelle la fonction `sizetab()` qui prend pour seul argument le pointeur sur le tableau, et renvoie le nombre d'éléments non négatifs contenus dans le tableau. Elle appelle ensuite la fonction `sumtab()`, qui calcule la somme de tous les entiers non négatifs contenus dans le tableau. Elle effectue la division entière et renvoie le quotient au programme appelant.

```

#include <stdio.h>

/* variables globales initialisées */
int tab[] = {23, 7, 12, 513, -1} ;

/* programme principal */
void main(void) {
    int x = arimean(tab);
    printf("%d", x);
    exit();
}

/* cette fonction renvoie la moyenne arithmétique des éléments d'un tableau */
int arimean(int t[]) {
    int n = sizetab(t);
    int x = sumtab(t);
    return (x / n);
}

/* cette fonction renvoie le nombre d'éléments d'un tableau */
int sizetab(int t[]) {
    int index = 0;
    while (t[index] >= 0) {
        index += 1;
    }
    return index;
}

/* cette fonction renvoie la somme des éléments d'un tableau */
int sumtab(int t[]) {
    int accu = 0;
    int index = 0;
    while (t[index] >= 0) {
        accu = accu + t[index];
        index += 1;
    }
    return accu;
}

```

Question 1

Donnez le code de la fonction arimean.

Solution:

```

1  # fonction arimean
2  # nr = 0 + $31 ; nv = 2 ; na = 1
3  arimean:
4      # prologue
5      addiu $29, $29, -16 # decrementation pointeur de pile
6      sw    $31, 12($29)  # sauvegarde de l'adresse de retour
7      sw    $4, 16($29)   # sauvegarde de l'argument de arimean
8      # corps de la fonction
9      jal   sizetab       # branchement a la fonction sizetab
10     sw    $2, 4($29)    # sauvegarde du resultat dans la variable n dans la pile

```

```

11     lw      $4,      16($29)      # recuperation de l'argument (sauve dans le prologue)
12     jal     sumtab              # branchement a la fonction sumtab
13     sw      $2,      8($29)      # ecriture x
14     lw      $8,      4($29)      # recuperation de n
15     div     $2,      $8          # division entiere
16     mflo    $2                # stockage resultat dans $2
17     # epilogue
18     lw      $31,     12($29)      # restauration adresse de retour
19     addiu   $29,     $29,      16 # incrementation pointeur de pile
20     jr      $31                # retour au programme appelant

```

Question 2

Donnez le code des fonction sumtab et sizetab.

Solution:

```

1  # fonction sizetab
2  # nr = 1 ($31) ; nv = 1 ; na = 0
3  # index : $2 parce que c'est une fonction terminale donc $2 ne sera pas modifie par des appels
4  # on utilisera $4 directement pour le parcours des cases du tableau
5  sizetab:
6      # prologue
7      addiu   $29,     $29,      -8 # decrementation pointeur de pile
8      sw      $31,     4($29)      # sauvegarde de l'adresse de retour
9      # corps de la fonction
10     ori     $2,      $0,        0 # initialisation index
11
12 loop0:
13     sll     $8,      $2,        2 # index * 4
14     addu    $8,      $4,        $8 # &t[index]
15     lw      $8,      0($8)        # chargement tab[i] dans $9
16     bltz    $8,      fin_loop0    # test de sortie de boucle
17     addiu   $2,      $2,        1 # incrementation index
18     j       loop0                # retour test sortie de boucle
19 fin_loop0:
20     # epilogue
21     lw      $31,     4($29)      # restauration de l'adresse de retour
22     addiu   $29,     $29,      8  # incrementation pointeur de pile
23     jr      $31                # retour au programme appelant
24
25 # fonction sumtab
26 # on choisit $2 pour accumulateur
27 # on choisit $8 pour index
28 # nr = 0 + $31 + nv = 2 + na = 0
29 # prologue
30 sumtab:
31     addiu   $29,     $29,      -12
32     sw      $31,     8($29)
33     ori     $2,      $0,        0 # initialisation registre accumulateur
34     addiu   $8,      $0,        0 # initialisation index
35 loop1:
36     sll     $9,      $8,        2 # index * 4
37     addu    $9,      $9,      $4  # &t[index]
38     lw      $9,      0($9)        # chargement tab[i] dans $9
39     bltz    $9,      fin_loop1    # test de sortie de boucle
40     addu    $2,      $2,      $9  # accumulation
41     addiu   $8,      $8,        1 # incrementation index
42     j       loop1

```

```

43 fin_loop1:                                # retour test sortie boucle
44     lw      $31,    8($29)
45     addiu   $29,    $29,    12
46     jr      $31                                # retour au programme appelant

```

Question 3

Donnez le code du programme principal ainsi que les déclarations des variables globales du programme.

Solution:

```

1  # variables globales initialisees
2  .data
3      tab: .word 23, 7, 12, 513, -1
4
5
6  .text
7  # programme principal
8  # nv = 1 na = 1
9  main:
10     # prologue
11     # na = 1, nr = 0, nv = 1
12     addiu   $29,    $29,    -8 # decrementation pointeur de pile $29
13     # corps de la fonction
14     lui     $4,     0x1001     # chargement adresse de tab dans $4
15     jal     arimean           # branchement a la fonction mean
16     sw      $2,     4($29)     # ecriture x
17     lw      $4,     4($29)     # argument de l'appel systeme affichage
18     ori     $2,     $0,     1  # code de l'appel systeme affichage d'entier
19     syscall
20
21     # epilogue
22     addiu   $29,    $29,     8  # decrementation pointeur de pile $29
23     ori     $2,     $0,    10   # code de l'appel systeme 'exit'
24     syscall

```

Exercice 3 – Élément de la suite de Fibonnaci

Question 1

Écrivez le code assembleur correspondant au code C ci-dessous. Ce code contient une fonction calculant le nombre de Fibonnaci à l'ordre n et un programme principal affichant le résultat pour une valeur stockée dans une variable globale.

```

int n = 4;

void main() {
    printf("%d", fib(n));
    exit();
}

int fib(int n) {
    if (n == 0 || n == 1) {
        return 1;
    }
}

```

```

    }
    else {
        return fib(n - 1) + fib(n - 2);
    }
}

```

Solution:

```

1  .data
2  n: .word 4
3
4  .text
5  main:
6      addiu $29, $29, -4      # na = 1, nv = 0
7      lui   $8, 0x1001
8      lw    $4, 0($8)        # 1er arg = n
9      jal   fib
10     ori   $4, $2, 0         # instruction d'adresse @a
11     ori   $2, $0, 1         # affichage du resultat
12     syscall
13
14     addiu $29, $29, 4
15     ori   $2, $0, 10
16     syscall
17
18
19  fib:
20     addiu $29, $29, -12     # 3 mots : na = 1 + nv = 0 + nr = 1 ($16) + $31
21     sw    $31, 8($29)
22     sw    $16, 4($29)
23     sw    $4, 12($29)      # on conserve n/1er param dans son emplacement
24
25     beq   $4, $0, _then    # cas n = 0
26     ori   $8, $0, 1
27     beq   $4, $8, _then    # cas n = 1
28     j     _else
29  _then:
30     ori   $2, $0, 1
31     j     _fin_if
32  _else:
33     addiu $4, $4, -1        # n - 1
34     jal   fib
35     ori   $16, $2, 0        # fib(n-1) dans reg persistant, inst. @b
36     lw    $4, 12($29)      # relecture n
37     addiu $4, $4, -2        # n - 2
38     jal   fib
39     addu  $2, $16, $2        # fib(n-1) + fib(n-2), inst. @c
40  _fin_if:
41     lw    $31, 8($29)
42     lw    $16, 4($29)
43     addiu $29, $29, 12
44     jr    $31

```

Question 2 – Représentation graphique de l'arbre d'appels

Représentez graphiquement l'arbre des appels engendrés par l'appel `fib(n)` avec `n` qui vaut 4.

Solution:

```

fib(4) -> fib(3) -> fib(2) -> fib(1)
                    -> fib(0)
                -> fib(1)
            -> fib(2) -> fib(1)
                -> fib(0)

```

Question 3 – Représentation graphique de la pile

Numérotez les instructions de votre code assembleur.

Représentez graphiquement l'évolution de la pile au cours de l'exécution. Vous indiquerez par @i les adresses de retour que vous mettrez sur la pile, avec i correspondant au numéro de l'instruction associée à cette adresse de retour.

Solution:

L'évolution de la pile est représentée dans la Figure 2. On se limite ici à fib(3).




FIGURE 2 – Evolution de la pile lors d'un appel à fib(3)